Sample solutions for Problem 5 in Problem Set 13 with `Macaulay2`

Chris Peterson and Hirotachi Abo

Problem 5 (Set 13). Let k be an algebraically closed field, let $F = y^2 - x^3 - x^2$ and $G = (x^2+y^2)^2 + 3x^2y - y^3$ be polynomials in $R := k[x,y]$, and let C_1 and C_2 denote the curves $V(F)$ and $V(G)$ in $\mathbb{A}^2$ respectively. Then the point $P = (0,0)$ is contained in both C_1 and C_2. We compute the intersection number of $I(P, F \cap G)$. Recall that $I(P, F\cap G)$ is defined to be the following vector space

$$\dim_k(\mathcal{O}_P(\mathbb{A}^2)/(F,G)\mathcal{O}_P(\mathbb{A}^2)),$$

where $\mathcal{O}_P(\mathbb{A}^2)$ is the local ring of $\mathbb{A}^2$ at P.

Proposition. Let F and G be polynomials in R. Suppose that F and G have no common factor and that $P \in V(F) \cap V(G)$. Then

$$I(P, F\cap G) = \lim_{n\to\infty} \dim_k \left(R/(I(P)^n, F, G)\right).$$

Proof. Without loss of generality, we may assume that $P = (0,0)$. So $I(P) = (x,y)$, and thus

$$I(P, F \cap G) = \dim_k \left(R_{(x,y)}/(F,G)_{(x,y)}\right).$$

Let $I_0 \cap \cdots \cap I_m$ be the primary decomposition of (F,G). Assume that $V(I_0) = P$. Then $(F,G)_{(x,y)} = (I_0)_{(x,y)}$, because any element of $R \setminus I_0$ correspond to a unit in $R_{(x,y)}$. This implies that

$$\dim_k(R_{(x,y)}/(F,G)_{(x,y)}) = \dim_k(R/I_0)_{(x,y)} = \dim_k R/I_0.$$

Consider the ideal $I_n = I(P)^n + (F,G)$. This ideal is equal to

$$(I(P)^n + I_0) \cap (I(P)^n + I'),$$

where $I' = I_1 \cap \cdots \cap I_m$. Since $V(I(P)^n + I') = \emptyset$, we obtain $I(P)^n + I' = k[x,y]$ (weak Nullstellensatz). So $I_n = I(P)^n + I_0$. Since $V(I_0) = P$, the $\sqrt{I_0} = (x,y)$. This implies that there exists an $n_0 \in \mathbb{N}$ such that for all $n \geq n_0$, $I(P)^n \subseteq I_0$. For such an n,

$$\dim_k(R/(I(P)^n, F, G)) = \dim_k(R/I_0),$$

which completes the proof. $\square$

For each n, the quotient ring $Q_n := R/(I(P)^n, F, G)$ is a finite-dimensional vector space over k. By using the gröbner basis technique, we can therefore describe the monomial basis B_n for this vector space (Macaulay's theorem), and the dimension of Q_n can be determined by counting the number of vectors in B_n. In `Macaulay2`, we can compute the monomial basis M for a given ring with the command `basis` if the ring is a finite-dimensional vector space. Let l be the number of the monomials in M. Then M is represented by a $1 \times l$ matrix:

$$Q_n \overset{M}{\leftarrow} Q_n^{\oplus l} = \overbrace{Q_n \oplus \cdots \oplus Q_n}^{l} .$$

The number of columns of this matrix is equal to the number of generators of the source. The source of a given matrix is represent by `source` plus the name of the matrix. The number of generators of the source can be counted with `numgens`.

Let a_n denote $\dim_k(Q_n)$. To determine the limit $\lim_{n\to\infty} a_n$, compute a_n for each $n \in \mathbb{N}$, and then terminate when $a_n = a_{n+1}$ (note that $a_k \leq a_{k+1}$ for each $k \in \mathbb{N}$). This algorithm can be implemented in `Macaulay2`. Let $I(P)$ be the ideal of the point $P = (0, 0)$, and let F and G be irreducible polynomials in $k[x, y]$ with no common factors. The function `intersectionNumber` takes as an input the triple of $I(P)$, (F) and (G), and returns the intersection number $I(P, F \cap G)$ at P. Here is the code of the function `intersectionNumber`:

```
i1 : intersectionNumber=(point,idl,idl')->(
          i:=0;
          isStable:=false;
          while not isStable do (
          d:=numgens source basis(ringA2/(idl+idl'+point^(i+1)));
          d':=numgens source basis(ringA2/(idl+idl'+point^(i+2)));
          if d===d' then (
             isStable=true
             );
          i=i+1;
          );
          d)

o1 = intersectionNumber

o1 : Function
```

Compute the intersection number $I(P, F \cap G)$ of C_1 and C_2 at $P = (0, 0)$:

```
i2 : KK=QQ
```

```
o2 = QQ

o2 : Ring

  --  the class of all rational numbers

i3 : ringA2=KK[x,y]

o3 = ringA2

o3 : PolynomialRing

i4 : F=ideal(y^2-x^3-x^2)

               3    2    2
o4 = ideal(- x  - x  + y )

o4 : Ideal of ringA2

i5 : G=ideal((x^2+y^2)^2+2*x^2*y-y^3)

             4     2 2    4     2     3
o5 = ideal(x  + 2x y  + y  + 2x y - y )

o5 : Ideal of ringA2

i6 : P=ideal(x,y)

o6 = ideal (x, y)

o6 : Ideal of ringA2

i7 : intersectionNumber(P,F,G)

o7 = 6
```

So $I(P, F \cap G) = 6$.

The proof of the above proposition tells us that if $I_0 \cap I_1 \cap \cdots \cap I_m$ is the primary decomposition of (F, G) and if $V(I_0) = P$, then $I(P, F \cap G) = \dim_k(R/I_0)$. Fortunately, we know the ideal $I(P)$ of P. So we can take

$V(I_0)$ away from $V(F,G)$ by computing $(F,G) : I(P)^\infty$, which is the ideal $I' = I_1 \cap \cdots \cap I_m$. Then take the ideal quotient $(F,G) : I'$. This gives us the ideal I_0:

```
i8 : I'=saturate(F+G,P)

             3     2            2                    2    2            2       2
o8 = ideal (y  + 4x  - 8x*y - y  + x - 12y - 3, x*y  + x  - x*y + 3y  + y, x y -

o8 : Ideal of ringA2

i9 : I0=(F+G):I'

             3     2    2    2   2    3    2    2
o9 = ideal (y , x*y  + x  - y , x y, x  + x  - y )

o9 : Ideal of ringA2

i10 : numgens source basis(ringA2/I0)

o10 = 6
```